/*-----------------------------------------------------------------------------
	The contents of this file are subject to the Mozilla Public License
	Version 1.1 (the "License"); you may not use this file except in
	compliance with the License. You may obtain a copy of the License at
	http://www.mozilla.org/MPL/
	
	Software distributed under the License is distributed on an "AS IS"
	basis, WITHOUT WARRANTY OF ANY KIND, either express or implied. See the
	License for the specific language governing rights and limitations
	under the License.
	
	The Original Code is Fever Framework code.
	
	The Initial Developer of the Original Code is Romain Ecarnot.
	Portions created by Initial Developer are Copyright (C) 2006
	the Initial Developer. All Rights Reserved.
	
	Contributor(s):
	
	Unless required by applicable law or agreed to in writing, software
	distributed under the License is distributed on an "AS IS" BASIS,
	WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
	See the License for the specific language governing permissions and
	limitations under the License.
-----------------------------------------------------------------------------*/

import org.aswing.ASColor;
import org.aswing.JFrame;

import fever.display.ModalScreen;

/**
 * Modal frame component.
 *
 * @author Romain Ecarnot
 */
class fvaswing.components.FvModalFrame extends JFrame
{
	//-------------------------------------------------------------------------
	// Private properties
	//-------------------------------------------------------------------------
	
	private var _modal : ModalScreen;
	
	
	//-------------------------------------------------------------------------
	// Public API
	//-------------------------------------------------------------------------
	 
	/**
	 * Constructor.
	 */
	public function FvModalFrame( caption : String, modalScreen : ModalScreen )
	{
		super( modalScreen.getModalTarget(), caption, false );
		
		_modal = modalScreen;
	}
	
	/**
	 * Indicates if we use Blur filter on modal screen.<br />
	 * Visible when screenshot property is on.
	 */
	public function setBlurFilterEnabled( b : Boolean ) : Void
	{
		if( b != _modal.enableBlurEffect )
		{
			_modal.enableBlurEffect = b;
			_modal.update();
		}	
	}
	
	/**
	 * Returns {@code true} if blur filter is activate.
	 */
	public function getBlurFilterEnabled() : Boolean
	{
		return _modal.enableBlurEffect;	
	}
	
	/**
	 * Sets modal screen background color
	 */
	public function setModalBackground( color : ASColor ) : Void 
	{
		_modal.setColor( color.getRGB() );
		_modal.setAlpha( color.getAlpha() );
		_modal.update();
	}
	
	/**
	 * Sets modal screen background color
	 */
	public function getModalBackground() : ASColor 
	{
		return new ASColor( _modal.getColor(), _modal.getAlpha() );
	}
	
	/**
	 * Overrides for modalscreen managment.
	 */
	public function show() : Void
	{
		_modal.update();
		_modal.show();
		
		super.show();	
	}
	
	/**
	 * Overrides for modalscreen managment.
	 */
	public function tryToClose() : Void
	{
		if( defaultCloseOperation == HIDE_ON_CLOSE )
		{
			_modal.hide();
			hide();
		}
		else if( defaultCloseOperation == DISPOSE_ON_CLOSE )
		{
			_modal.dispose();
			dispose();
		}		
	}
	
		
}